Answer Set Programming

Diagnosis

Martin Baláž

Department of Applied Informatics
Comenius University in Bratislava

2009/2010

Outline

Diagnostic Problem

Definition (Diagnostic Problem)

A *diagnostic problem* is a triple $\langle H, T, O \rangle$ where

- H is a set of ground atoms and is referred to as *hypothesis*
- T is a *theory* describing a system
- O is a set of ground literals and is referred to as *observations*

A possible *diagnosis* is a subset $\Delta \subseteq H$.

Which diagnosis are acceptable?

- Generic Diagnosis
- Single Error Diagnosis
- Minimal Subset Diagnosis

Abductive Diagnosis

Definition (Abductive Diagnostic Problem)

An *abductive diagnostic problem* $\mathcal{P}$ is a triple $\langle H, T, O\rangle$ where

- H is a set of ground atoms
- T is a disjunctive normal logic program
- O is a set of ground literals

Definition (Abductive Diagnosis)

Let $\mathcal{P} = \langle H, T, O\rangle$ be an abductive diagnostic problem.
An *abductive diagnosis* is a set $\Delta \subseteq H$ such that $T \cup \Delta \models O$, i.e. there exists a stable model M of $T \cup \Delta$ such that $M \models O$.

Example



Example

$$H = \{broken\_bulb, power\_off, high\_curr\}$$

```
light_off :- power_off.
light_off :- broken_bulb.
light_off :- melted_fuse1, melted_fuse2.
melted_fuse1 v melted_fuse2 :- high_curr.
hot_plate1 :- not melted_fuse1, not power_off.
hot_plate2 :- not melted_fuse2, not power_off.
```

$$O = \{light\_off, \sim hot\_plate_1\}$$

$$\begin{aligned}\Delta_1 &= \{power\_off\}\\ \Delta_2 &= \{broken\_bulb\}\\ \Delta_3 &= \{broken\_bulb, high\_curr\}\end{aligned}$$

Consistency-Based Diagnosis

Definition (Consistency-Based Diagnostic Problem)

A *consistency-based diagnostic problem* $\mathcal{P}$ is a triple $\langle H, T, O \rangle$ where

- H is a set of ground atoms with predicate name *ab*
- T is a set of first-order sentences
- O is a set of ground literals

Definition (Consistency-Based Diagnosis)

Let $\mathcal{P} = \langle H, T, O \rangle$ be a consistency-based diagnostic problem. A *consistency-based diagnosis* is a set $\Delta \subseteq H$ such that $T \cup O \cup \Delta \cup \{\sim h \mid h \in H \setminus \Delta\}$ is consistent, i.e. there exists a model of $T \cup O \cup \Delta \cup \{\sim h \mid h \in H \setminus \Delta\}$.

Example



Example

```
out(X, 1) :- and(X), in1(X, 1), in2(X, 1), not ab(X).
out(X, 0) :- and(X), in1(X, 0), not ab(X).
out(X, 0) :- and(X), in2(X, 0), not ab(X).

out(X, 0) :- or(X), in1(X, 0), in2(X, 0), not ab(X).
out(X, 1) :- or(X), in1(X, 1), not ab(X).
out(X, 1) :- or(X), in2(X, 1), not ab(X).

out(X, 1) :- xor(X), in1(X, 1), in2(X, 0), not ab(X).
out(X, 1) :- xor(X), in1(X, 0), in2(X, 1), not ab(X).
out(X, 0) :- xor(X), in1(X, Y), in2(X, Y), not ab(X).
```

Example

```
gate(X) :- and(X).
gate(X) :- or(X).
gate(X) :- xor(X).

in1(X, 0) v in1(X, 1) :- gate(X).
in2(X, 0) v in2(X, 1) :- gate(X).
out(X, 0) v out(X, 1) :- gate(X).

:- out(X, 0), out(X, 1).
:- in1(X, 0), in1(X, 1).
:- in2(X, 0), in2(X, 1).
```

Example

```
in1(a1, S) :- in1(x1, S).
in1(x1, S) :- in1(a1, S).

in2(a1, S) :- in2(x1, S).
in2(x1, S) :- in2(a1, S).

in2(x2, S) :- in1(a2, S).
in1(a2, S) :- in2(x2, S).

out(x1, S) :- in1(x2, S).
in1(x2, S) :- out(x1, S).

out(x1, S) :- in2(a2, S).
in2(a2, S) :- out(x1, S).

out(a1, S) :- in2(o1, S).
in2(o1, S) :- out(a1, S).

out(a2, S) :- in1(o1, S).
in1(o1, S) :- out(a2, S).
```

Example

$$H = \{ab(x_1), ab(x_2), ab(a_1), ab(a_2), ab(o_1)\}$$

$$O = \{in_1(x_1, 1), in_2(x_1, 0), in_1(a_2, 1), out(x_2, 1), out(o_1, 0)\}$$

$$\Delta_1 = \{ab(x_1)\}$$
$$\Delta_2 = \{ab(x_2), ab(o_1)\}$$
$$\Delta_3 = \{ab(x_2), ab(a_2)\}$$